Affine Abbildungen 2d

Mathematik in wxMaxima www.mathematik-verstehen.de Haftendorn Jan 2011

0.1 Handling

0.2 Dieses ist das Einführungsbeispiel

0.3 Inhalt

Figure 1:

1 Urbild

1.1 Definition

Urbild

```
(%i18) myUr:transpose(matrix([2,0],[2,1],[1,1],[1,3],[0,3],[0,0],[2,0],[1,1/2]));
```

(%o18) $\begin{bmatrix} 2 & 2 & 1 & 1 & 0 & 0 & 2 & 1 \\ 0 & 1 & 1 & 3 & 3 & 0 & 0 & \frac{1}{2} \end{bmatrix}$

1.2 Zeichnen

```
(%i19) load(draw)$
```

```
(%i20) xmin:0$ xmax:7 $ ymin:0$ ymax:5$
```

```
(%i24) urbild:gr2d(xrange = [xmin,xmax], yrange = [ymin,ymax],points_joined = true,color=green,
            line_width    = 7,  point_size    = 1, point_type    = up_triangle,
            grid=true,points(myUr)   )$
```

```
--> draw(urbild)$
```

Figure 2:



2 Abbildung

2.1 Abbildung der Einheitsvektoren

```
(%i1) A:transpose(matrix([3/2,1/2],[1/2,3/4]));
```

(%o1) $\begin{bmatrix} \frac{3}{2} & \frac{1}{2} \\ \frac{1}{2} & \frac{3}{4} \end{bmatrix}$

```
(%i25) startbild:gr2d(xrange = [xmin,xmax], yrange = [ymin,ymax],
                points_joined = true,color=green,
                line_width    = 7,  point_size    = 1, point_type    = up_triangle,
                grid=true,points(myUr) ,
                grid=true, line_width=2,point_size=1, head_length=0.2,
                color=blue,vector([0,0],[A[1,1],A[2,1]]),
                color=black,vector([0,0],[A[1,2],A[2,2]]))$
```

```
--> draw(startbild)$
```

Figure 3:



Der blaue   Vektor ist das Bild des x-Einheitsvektors.
Der schwarze Vektor ist das Bild des y-Einheitsvektors.

2.2 Abbildungsgleichung für einen Punkt

Definition der Translation

(%i2) tx:3$ ty:2$

(%i4) tv:transpose(matrix([tx,ty]));

(%o4) $\begin{bmatrix} 3 \\ 2 \end{bmatrix}$

Allgemeine affine Abbildung

(%i7) f(xv):=A.xv+tv;

(%o7) $\mathrm{f}(xv):=A\ .\ xv+tv$

(%i8) f(xv);

(%o8) $\begin{bmatrix} \frac{3}{2} & \frac{1}{2} \\ \frac{1}{2} & \frac{3}{4} \end{bmatrix} .\ xv+\begin{bmatrix} 3 \\ 2 \end{bmatrix}$

Abbildung eines beliebigen Punktes

(%i9) pv:transpose(matrix([px,py]));

(%o9) $\begin{bmatrix} px \\ py \end{bmatrix}$

(%i10) f(pv);

(%o10) $\begin{bmatrix} \frac{py}{2}+\frac{3\,px}{2}+3 \\ \frac{3\,py}{4}+\frac{px}{2}+2 \end{bmatrix}$

Abbildung eines konkreten Punktes

(%i11) pv:transpose(matrix([2,1]));

(%o11) $\begin{bmatrix} 2 \\ 1 \end{bmatrix}$

Nur affine Verzerrung

(%i26) A.pv;

(%o26) $\begin{bmatrix} \frac{7}{2} \\ \frac{7}{4} \end{bmatrix}$

gesamt

```
(%i27) f(pv);
       %,numer;
```

$$(\%o27)\quad \begin{bmatrix} \frac{13}{2} \\ \frac{15}{4} \end{bmatrix}$$

$$(\%o28)\quad \begin{bmatrix} 6.5 \\ 3.75 \end{bmatrix}$$

2.3 Affine Verzerrung für das ganze Urbild

```
(%i29) myUr;
       myABild:A.myUr;
       %,numer;
```

$$(\%o29)\quad \begin{bmatrix} 2 & 2 & 1 & 1 & 0 & 0 & 2 & 1 \\ 0 & 1 & 1 & 3 & 3 & 0 & 0 & \frac{1}{2} \end{bmatrix}$$

$$(\%o30)\quad \begin{bmatrix} 3 & \frac{7}{2} & 2 & 3 & \frac{3}{2} & 0 & 3 & \frac{7}{4} \\ 1 & \frac{7}{4} & \frac{5}{4} & \frac{11}{4} & \frac{9}{4} & 0 & 1 & \frac{7}{8} \end{bmatrix}$$

$$(\%o31)\quad \begin{bmatrix} 3 & 3.5 & 2 & 3 & 1.5 & 0 & 3 & 1.75 \\ 1 & 1.75 & 1.25 & 2.75 & 2.25 & 0 & 1 & 0.875 \end{bmatrix}$$

```
(%i32) ABild:gr2d(xrange = [xmin,xmax], yrange = [ymin,ymax],
                  points_joined = true,color=green,
                  line_width    = 7,  point_size    = 1, point_type    = up_triangle,
                  grid=true,points(myUr) ,
                  grid=true, line_width=2,point_size=1, head_length=0.2,
                  color=blue,vector([0,0],[A[1,1],A[2,1]]),
                  color=black,vector([0,0],[A[1,2],A[2,2]]),
                  line_width=7,point_size=1, color=red,
                  points(myABild)
       )$ draw(ABild)$
```

Figure 4:



2.4 Ganze affine Abbildung für das Urbild

Nun muss zu jedem dieser Bildpunkte der Translationsvektor addiert werden. Dazu muss man ihn passend "aufblähen" zu einer Transformationsmatrix.

```
(%i34) npk:length(transpose(myUr));
(%o34) 8
```

```
(%i35) mtv(tv):=block ([m],
                    m:tv,for i:1 thru npk-1 do ( m:addcol(m,tv)),return( m))$
```

```
(%i36) mtv(tv);
```

(%o36) $\begin{bmatrix} 3 & 3 & 3 & 3 & 3 & 3 & 3 & 3 \\ 2 & 2 & 2 & 2 & 2 & 2 & 2 & 2 \end{bmatrix}$

Gesamte Abbildung des Urbildes

```
(%i37) fm(myUr):=A.myUr+mtv(tv);
```

(%o37) $\mathrm{fm}(myUr):=A\,.\,myUr+\mathrm{mtv}(tv)$

```
(%i38) myBild:fm(myUr);
       %,numer;
```

(%o38) $\begin{bmatrix} 6 & \frac{13}{2} & 5 & 6 & \frac{9}{2} & 3 & 6 & \frac{19}{4} \\ 3 & \frac{15}{4} & \frac{13}{4} & \frac{19}{4} & \frac{17}{4} & 2 & 3 & \frac{23}{8} \end{bmatrix}$

(%o39) $\begin{bmatrix} 6 & 6.5 & 5 & 6 & 4.5 & 3 & 6 & 4.75 \\ 3 & 3.75 & 3.25 & 4.75 & 4.25 & 2 & 3 & 2.875 \end{bmatrix}$

```
(%i40) Bild:gr2d(xrange = [xmin,xmax], yrange = [ymin,ymax],
              points_joined = true,color=green,
              line_width    = 7,  point_size    = 1, point_type    = up_triangle,
              grid=true,points(myUr) ,
              grid=true, line_width=2,point_size=1, head_length=0.2,
              color=blue,vector([0,0],[A[1,1],A[2,1]]),
              color=black,vector([0,0],[A[1,2],A[2,2]]),
              line_width=2,point_size=1, color=pink,
              points(myABild) ,
               color=red,vector([0,0],[tv[1,1],tv[2,1]]),
              line_width=7,color=red,
              points(myBild)
       )$  draw(Bild)$
```

Figure 5:



3 Eigenwerte und Eigenvektoren

3.1 Vorgefertigte Beschaffung

(%i42) eigenvalues(A);

(%o42) $[[\frac{7}{4},\frac{1}{2}],[1,1]]$

(%i43) ev_all:eigenvectors(A);

(%o43) $[[[\frac{7}{4},\frac{1}{2}],[1,1]],[[[1,\frac{1}{2}]],[[1,-2]]]]$

Die Liste ist so zu deuten:
Erste Unterliste: die beiden Eigenwerte, dann ihre Vielfachheiten.
Zweite Unterliste: erster Eigenvektor, zweiter Eigenvektor

3.2 Beschaffung von Hand

(%i44) A;

(%o44) $\begin{bmatrix} \frac{3}{2} & \frac{1}{2} \\ \frac{1}{2} & \frac{3}{4} \end{bmatrix}$

(%i45) E: matrix([1,0], [0,1]);

(%o45) $\begin{bmatrix} 1 & 0 \\ 0 & 1 \end{bmatrix}$

(%i46) A-x*E;

(%o46) $\begin{bmatrix} \frac{3}{2}-x & \frac{1}{2} \\ \frac{1}{2} & \frac{3}{4}-x \end{bmatrix}$

(%i47) determinant(A-x*E);

(%o47) $\left(\frac{3}{4}-x\right)\left(\frac{3}{2}-x\right)-\frac{1}{4}$

Auch dieses geht direkt

--> charpoly(A,x);

(%o42) $\left(\frac{3}{4}-x\right)\left(\frac{3}{2}-x\right)-\frac{1}{4}$

(%i48) solve(charpoly(A,x)=0,x);

(%o48) $[x=\frac{7}{4},x=\frac{1}{2}]$

Hier sieht man die beiden Eigenwerte.

3.3 Darstellung, geometrische Bedeutung

(%i49) ew:ev_all[1][1]; ev1:ev_all[2][1][1];ev2:ev_all[2][2][1];

(%o49) $[\frac{7}{4},\frac{1}{2}]$

(%o50) $[1,\frac{1}{2}]$

(%o51) $[1,-2]$

```
(%i54) evBild:gr2d(xrange        = [-1,7], yrange        = [-2,5],
                points_joined = true,color=green, line_width=2,
                points(myUr),
                grid=true, line_width=2,point_size=1, head_length=0.1,
                line_type=dots,  color=orange,
                vector([0,0],[ev1[1],ev1[2]]),
                color=red,vector([0,0],[ev2[1],ev2[2]]),head_length=0.2,
                color=orange,
                vector([0,0],ew[1]*[ev1[1],ev1[2]]),
                color=red,vector([0,0],ew[2]*[ev2[1],ev2[2]]),
                line_width=2,point_size=1, color=red,
                points(myABild)
                )$  draw(evBild)$
```

Figure 6:



Die großen Pfeilspitzen bezeichnen die Bilder der Eigenvektoren.

Bei dem Einführungsbeispiel sieht man, dass der eine Eigenwert seinen Eigenvektor (orange) knapp verdoppelt *7/4, der andere halbiert ihn (1/2).

4 Wirkung einer Linearen Abbildung

4.1 Urbild des Einheitskreises?

(%i56) A;

(%o56) $\begin{bmatrix} \frac{3}{2} & \frac{1}{2} \\ \frac{1}{2} & \frac{3}{4} \end{bmatrix}$

(%i57) S:A.A;

(%o57) $\begin{bmatrix} \frac{5}{2} & \frac{9}{8} \\ \frac{9}{8} & \frac{13}{16} \end{bmatrix}$

(%i61) pt:matrix([x,y]); p:transpose(pt);ellipseS:facsum(pt.S.p=1);

(%o61) $\begin{bmatrix} x & y \end{bmatrix}$

(%o62) $\begin{bmatrix} x \\ y \end{bmatrix}$

(%o63) $\frac{13\, y^2 + 36\, x\, y + 40\, x^2}{16} = 1$

```
(%i91) draw(gr2d(line_width=3, color=red,grid=true,
            implicit(ellipseS, x, -1.5,1.5, y, -2,2),color=blue,
            implicit(x^2+y^2=1, x, -1.5,1.5, y, -2,2),

        grid=true, line_width=2,point_size=1, head_length=0.1,
            line_type=dots,  color=orange,
            vector([0,0],[ev1[1],ev1[2]]/sqrt(5/4)),
            color=red,vector([0,0],[ev2[1],ev2[2]]/sqrt(5)),head_length=0.2,
            color=orange,
            vector([0,0],ew[1]^-1*[ev1[1],ev1[2]]/sqrt(5/4)),
            color=red,vector([0,0],ew[2]^-1*[ev2[1],ev2[2]]/sqrt(5))

           ))$
```

Figure 7:



--> charpoly(S,x);

(%o50) $\left(\frac{13}{16}-x\right)\left(\frac{5}{2}-x\right)-\frac{81}{64}$

--> evs:eigenvectors(S);

(%o51) $[[[\frac{49}{16},\frac{1}{4}],[1,1]],[[[1,\frac{1}{2}]],[[1,-2]]]]$

Dies passt zur Thoerie bei symmetrischem A: Eigenwerte quadriert, dieselben Eigenvektoren

4.2 Transformationsmatrix

--> evs[2][1][1];

(%o52) $[1,\frac{1}{2}]$

```
--> Tt:matrix(evs[2][1][1]/sqrt(5/4),evs[2][2][1]/sqrt(5));T:transpose(Tt)$
```

(%o76) $\begin{bmatrix} \frac{2}{\sqrt{5}} & \frac{1}{\sqrt{5}} \\ \frac{1}{\sqrt{5}} & -\frac{2}{\sqrt{5}} \end{bmatrix}$

```
--> facsum(T.S.Tt);
```

(%o78) $\begin{bmatrix} 2 & -1 \\ -1 & 1 \end{bmatrix}$

Diesen Schritt braucht man nicht wirklich zu machen, denn die Theorie garantiert hier eine Diagonalmatrix mit den Eigenwerten in der Hauptdiagonalen.

```
--> elliS:(49/16*x^2+1/4*y^2=1);
```

(%o56) $\frac{y^2}{4}+\frac{49\,x^2}{16}=1$

4.3 Bild des Einheitskreises

Auf welche Punkte wird der Einheitskreis abgebildet?

```
(%i67) A^^-1;
```

(%o67) $\begin{bmatrix} \frac{6}{7} & -\frac{4}{7} \\ -\frac{4}{7} & \frac{12}{7} \end{bmatrix}$

Die Inverse einer symmetrischen Matrix ist symmetrisch

```
(%i68) Q:A^^-2;
```

(%o68) $\begin{bmatrix} \frac{52}{49} & -\frac{72}{49} \\ -\frac{72}{49} & \frac{160}{49} \end{bmatrix}$

```
(%i72) ellipseQ:facsum(pt.Q.p=1);
```

(%o72) $\frac{4\,(40\,y^2-36\,x\,y+13\,x^2)}{49}=1$

(%i90) draw(gr2d(line_width=3, color=red,grid=true,
implicit(ellipseQ, x, -2,2, y, -2,2),color=blue,
implicit(x^2+y^2=1, x, -1.5,1.5, y, -2,2),

grid=true, line_width=2,point_size=1, head_length=0.1,
line_type=dots, color=orange,
vector([0,0],[ev1[1],ev1[2]]/sqrt(5/4)),
color=red,vector([0,0],[ev2[1],ev2[2]]/sqrt(5)),head_length=0.2,
color=orange,
vector([0,0],ew[1]*[ev1[1],ev1[2]]/sqrt(5/4)),
color=red,vector([0,0],ew[2]*[ev2[1],ev2[2]]/sqrt(5))

))$

Figure 8:



(%i92) charpoly(Q,x);

(%o92) $\left(\frac{52}{49}-x\right)\left(\frac{160}{49}-x\right)-\frac{5184}{2401}$

(%i93) evs:eigenvectors(Q);

(%o93) $[[[\frac{16}{49},4],[1,1]],[[[1,\frac{1}{2}]],[[1,-2]]]]$

Dies passt zur Thoerie bei symmetrischem A: Kehrwerte der Eigenwerte quadriert,dieselben Eigenvektoren